GAMEBOY ASM HACKING:

LE COMPRESSIONI

Written by ev0 aka DarkDreAm

ev0luzione@yahoo.it
Questo documento si rivolge a chi ha un minimo di conoscenze di ASM su gameboy, in caso contrario correte a studiare le basi di questo processore molto semplice e simile a uno z80…

Comunque usate questa guida a vostro rischio e pericolo :)

Io vi guiderò solo nell’utilizzo dei tool e vi spiegherò come affrontare una compressione su gameboy.

Avrete bisogno del Documento di Pan/Anthrox possibilmente quello più aggiornato reperibile sul sito http://www.work.de/nocash/gmb.html, una bella mappa di memoria e qualche altro documento utile come la lista degli opcode e le loro funzioni.

Dovreste anche avere delle basi sul funzionamento delle compressioni.

Alcuni possono essere reperiti su devrs.com, gli altri, cercateli con google :)

Per il resto ci serve:

IDA Pro (vi ricordo che c’è una versione Free)

NO$GMB

Editor Esadecimale

La rom presa in esame è R-Type 2.

Ha una compressione alquanto strana e semplice da comprendere, ha qualcosa a che fare con le compressioni LZ, ma termina con un EOF code $FFFF come le RLE…

Questo è già un mio progetto… quindi evitate di utilizzare la “famosa” tecnica del copia&incolla mettendo il vostro nome su un progetto di traduzione.

Apriamo il nostro fido NO$GM Debugger, vi consiglio di settare nelle opzioni di emulazione di eseguire il gioco nella piccola finestra nel debugscreen.

Apriamo Window, Tile Viewer.

Adesso apriamo la rom, e magicamente vedremo apparire le font decompresse e il resto della grafica nella VRAM…

Sappiamo che la VRAM ha 2 “banchi” di tile selezionabili, uno a $8000 e uno a $8800 che si intersecano nelle tiles $80-$FF…

Cosa stiamo a fare qui? A decomprimere la font, giusto?

Partiamo con gli spazi bianchi da $9000, settiamo un breakpoint (quando l’emulatore incorrerà in quella situazione si bloccherà nell’istruzione seguente a quella che ha causato il breakpoint), quindi Debug, Define Break Condition oppure direttamente CTRL+B.

Se leggiamo l’help del NO$GMB vediamo che un breakpoint, in questo caso, dobbiamo settarlo così: (90E0..90F0)!

Questo significa che l’emulatore si bloccherà subito dopo che un’istruzione ha scritto qualcosa nel range di memoria 90E0-90F0.

Adesso premiamo *, oppure Run, Reset & Run.

L’emulatore si è bloccato ohhhhh… ma guardiamo bene nel tile viewer (che vi consiglio di tenere sempre aperto e ben visibile sotto la finistra principale) e vedremo però, che le tile tra 90E0 e 90DF che dovrebbero essere state decompresse, non ci sono. Quindi premiamo qualche volta F9 per far “correre” (run xD) il gioco e vedremo le font che dovevano esserci. E tutto il resto che stava sopra a partire da $8800? Eheheh ci arriviamo dopo…

Bene adesso se vediamo il pc (il registro che tiene conto della posizione in cui siamo nella memoria) è diverso dalle istruzioni in cui eravamo incappati all’inizio…

Premiamo F9 altre due tre volte… hehe che bello stà decomprimendo la font… e la cosa più bella è che utilizza sempre le stesse due istruzioni xD siamo nel mezzo della routine di decompressione…

Facciamo sempre attenzione ai registri, in DE abbiamo $90XX ed è l’unico registro nel range della VRAM quindi sicuramente sarà usato come registro di destinazione. In hl c’è $40XX, l’unico nel range della rom quindi sarà il sorgente…

Saliamo un po’ su e tra le istruzioni incontriamo un ret z, quindi voi direte che la routine incomincia a $BD6… va bene apriamo il nostro fidato IDA Pro, apriamo il file e selezioniamo come processore z80: gb, premiamo G e andiamo all’indirizzo $BD6, premiamo C per far cominciare l’analisi… guardiamo un po’ più su… tho! Ha disassemblato a partire da $BCF, che significa? Significa che c’è qualche jmp a quell’indirizzo nella routine di decompressione…

Se aveste fatto attenzione però quello era un ret condizionale, il che significa che c’è la possibilità che il codice che stà dopo venga eseguito e che quello che stà tra il ret a $BCE e il ret z non fa altro che controllare che il byte in A sia uguale a quello successivo e che siano quindi tutte e due uguali a $FF, va bè poco importa, meno male che c’è IDA :)

Ma tutto questo non ci basta ;) Andiamo in NO$GMB (che non avete chiuso vero?) premiamo il tasto Home e l’emulatore ci porterà a $100 adesso premiamo CTRL+S e facciamo una ricerca per BCF. Dando uno sguardo al codice si intuisce che è come avevamo pensato, in hl c’è il puntatore alla grafica compressa nella rom, in de la destinazione :)

Quindi dobbiamo cercare qualcosa come:

ld hl, xxxxxxx

ld de, xxxxxxx

call 0BCF

premiamo CTRL+C per continuare la ricerca, la prima ricerca non va bene, la seconda potrebbe essere ma non è quella, è di troppo superiore a quel $40XX visto prima… continuiamo fino a che la ricerca non ci porta a ROM0:287C… bene bene prendiamo IDA PRO e andiamo alla locazione sotto a quel jr 289C quindi con G noi andremo a 286E…

ROM:286E ; ---------------------------------------------------------------------------

ROM:286E                 ld      a, 7

ROM:2870                 ld      [byte_D07D], a

ROM:2873                 call    sub_579

ROM:2876                 ld      de, $9000

ROM:2879                 ld      hl, $4000

ROM:287C                 call    sub_BCF

ROM:287F                 ld      de, $5754

ROM:2882                 ld      hl, $9C00

ROM:2885                 call    sub_2A37

ROM:2888                 ld      b, $14

ROM:288A                 ld      de, $9E00

ROM:288D                 ld      hl, $3776

ROM:2890                 call    sub_5B1

ROM:2893                 ld      de, $9E20

ROM:2896                 ld      b, $14

ROM:2898                 call    sub_5B1

ROM:289B                 ret

ROM:289B ; ---------------------------------------------------------------------------

Quella call mi insospettisce voglio sapere a che serve…

Andiamo a $579:

ROM:0579 ; ¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦ S U B R O U T I N E ¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦

ROM:0579

ROM:0579

ROM:0579 sub_579:                                ; CODE XREF: ROM:2873
p

ROM:0579                 push    af

ROM:057A                 ld      a, [byte_D093]

ROM:057D                 ld      [byte_D00D], a

ROM:0580                 pop     af

ROM:0581                 ld      [byte_D093], a

ROM:0584                 ld      [byte_2000], a

ROM:0587                 ret

ROM:0587 ; End of function sub_579

ROM:0587

ROM:0587 ; ---------------------------------------------------------------------------
Cosa sarà mai? Niente di speciale, se andiamo in NO$GMB a $584 ci dirà ROM Select…

Già, significa che il nostro caro mapper seleziona il banco che stà in A… in questo caso 7 :)

Questo significa che la nostra grafica compressa stà a $4000 nel banco 7 quindi $1C000 nella rom.

ROM:2879                 ld      hl, $4000

Ma adesso abbiamo il punto di partenza, io voglio avere la fine  ^__^

Vediamo un po’ il codice…

ROM:0BCF sub_BCF:                                ; CODE XREF: sub_BCF+1B
j

ROM:0BCF                                         ; sub_BCF+1F
j ...

ROM:0BCF                 ldi     a, [hl]

ROM:0BD0                 cp      [hl]

ROM:0BD1                 jr      nz, loc_BD6

ROM:0BD3                 cp      $FF

ROM:0BD5                 ret     z

ROM:0BD6

Il salto condizionale a BD1 può saltare qui!

Allora:

Carica in A il byte che stà ad hl e incrementa hl

Confronta con il byte puntato da HL che quindi punterà al byte dopo

E se sono diversi salta a BD6 altrimenti vede se sono tutti e due $FF

Se sono diversi esci dalla routine di compressione (ret z)

Se no continua con BD6…

Semplice no? Vuol dire che il nostro EOF code è $FFFF :)

Come facciamo a ottenere la fine della stream?

Ragionate, se il flag z è settato (1) vuol dire che il ret z funziona ed esce dalla routine quindi basterà settare un breakpoint così:

BD5, z
Il che significa che si fermerà su BD5 se il flag z è settato :)

Diamo un’occhiata ad hl…

4E5B, quindi a 1CE5B ( (il byte a 1CE5B incluso ovviamente xD)

Abbiamo l’inizio, abbiamo la fine, prendiamo il fido editor esadecimale e dumpiamo…

Adesso a voi la scelta decomprimere e recomprimere?

È possibilissimo visto la semplicità dell’algoritmo, addirittura, io mi sono riscritto il decompressore in asm su gb xD

Ma visto che non conosco il C, e sto avendo difficoltà a scriverlo in Pascal, lo sto scrivendo in ASM su gameboy xD ma ci vuole una vita… la soluzione migliore è quella di ripuntare la grafica decompressa dove vogliamo noi oppure lasciargliela decomprimere e poi sovrascriverla con la nostra, meglio così per questioni di spazio…

A $1B000 c’è tanto spazio da utilizzare… Copiamoci la nostra font decompressa e modificata.

A $3D50 (per evitare casini con il rom swtiching) metteremo il nostro codice :)

Apriamo la rom modificata col NO$GMB, andiamo a $286E, steppiamo (F7) fino a $2876 e scriviamo direttamente sulla tastiera (si aprirà la finestra per inserire l’istruzione da assemblare :)

jp  3D50

nop

nop

nop

nop

nop

nop

segnamoci i byte che son cambiati e andiamo a sovrascriverli nella rom con l’editor esadecimale.

Premiamo F7 e ci ritroviamo a 3D50, cominciamo a scrivere:

ld de,9000

ld hl,4000

call 0bcf

ld a,6

ld (D07D),a

ld (1000),a (che poi cambieremo in EA7DD0, purtroppo è l’asm del NO$ che dà problemi)
call 0579

ld hl,7000

ld de,9000

ld bc,02D0

inc b

inc c

jp 3d77

di

call 0680

ldi a,(hl)

ld (de),a

ei

inc de

dec c

jr nz, 3d6f

dec b

jr nz, 3d6f

ld a,7

ld (1000),a (che poi cambieremo in EA7DD0, purtroppo è l’asm del NO$ che dà problemi)
call 0579

jp 2879
Quindi…

A $2876:
C3503D000000000000

A $35D0:

110090210040CDCF0B3E06EA7DD0CD790521007011009001D002040CC3773DF3CD80062A12FB130D20F50520F23E07EA7DD0CD7905C37928

Il gioco è fatto :)

Abbiamo la nostra font bella e decompressa pronta da utilizzare ;)

Per chi vuole i commenti:

jp  3D50 
;salta a 3D50 dove stiamo mettendo la nostra routine

ld de,9000

ld hl,4000

call 0bcf        ;decomprime la grafica originaria in VRAM

ld a,6

ld (D07D),a
;(non 1000)

call 0579
;cambia banco rom al 6° dove c’è la nostra fontazza :)

ld hl,7000
;la posizione della nostra font in memoria

ld de,9000
;VRAM dove mettiamo la nostra font

ld bc,02D0
;il numero di byte da copiare (la dimensione della font)

inc b

inc c

jp 3d77

di

call 0680
;aspetta per il Vblank

ldi a,(hl)
;e

ld (de),a
;copia i byte dal sorgente nella destinazione

ei

inc de

dec c

jr nz, 3d6f

dec b

jr nz, 3d6f
;fino a qui la routine non fa altro che copiare quello che gli abbiamo detto

ld a,7

ld (D07D),a
;(non 1000) anche qui ;)

call 0579
;cambia banco (ritorna all’originario 7°)

jp 2879
ritorna al punto da cui abbiamo saltato qui.
Appendice A: La routine di decompressione

HL: sorgente

DE: destinazione

ROM:0BCF ; ---------------------------------------------------------------------------

ROM:0BCF

ROM:0BCF loc_BCF:                                ; CODE XREF: ROM:0BEA_j

ROM:0BCF                                         ; ROM:0BEE_j

ROM:0BCF                 ldi     a, [hl]

ROM:0BD0                 cp      [hl]

ROM:0BD1                 jr      nz, loc_BD6

ROM:0BD3                 cp      $FF

ROM:0BD5                 ret     z

ROM:0BD6

ROM:0BD6 loc_BD6:                                ; CODE XREF: ROM:0BD1_j

ROM:0BD6                 ld      b, a

ROM:0BD7                 ldi     a, [hl]

ROM:0BD8                 ld      c, a

ROM:0BD9                 ldi     a, [hl]

ROM:0BDA                 ld      [byte_D026], a

ROM:0BDD                 push    bc

ROM:0BDE                 ld      a, b

ROM:0BDF                 call    sub_BF0

ROM:0BE2                 pop     bc

ROM:0BE3                 ld      a, c

ROM:0BE4                 call    sub_BF0

ROM:0BE7                 ld      a, d

ROM:0BE8                 cp      $98 ; 'ÿ'

ROM:0BEA                 jr      nz, loc_BCF

ROM:0BEC                 ld      d, $88 ; 'ê'

ROM:0BEE                 jr      loc_BCF

ROM:0BF0

ROM:0BF0 ; ¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦ S U B R O U T I N E ¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦¦

ROM:0BF0

ROM:0BF0

ROM:0BF0 sub_BF0:                                ; CODE XREF: ROM:0BDF_p

ROM:0BF0                                         ; ROM:0BE4_p

ROM:0BF0                 ld      c, a

ROM:0BF1                 ld      b, 8

ROM:0BF3

ROM:0BF3 loc_BF3:                                ; CODE XREF: sub_BF0+18_j

ROM:0BF3                 rrc     c

ROM:0BF5                 jr      c, loc_C01

ROM:0BF7                 ld      a, [byte_D026]

ROM:0BFA                 call    sub_680

ROM:0BFD                 ld      [de], a

ROM:0BFE                 inc     de

ROM:0BFF                 jr      loc_C07

ROM:0C01 ; ---------------------------------------------------------------------------

ROM:0C01

ROM:0C01 loc_C01:                                ; CODE XREF: sub_BF0+5_j

ROM:0C01                 ldi     a, [hl]

ROM:0C02                 call    sub_680

ROM:0C05                 ld      [de], a

ROM:0C06                 inc     de

ROM:0C07

ROM:0C07 loc_C07:                                ; CODE XREF: sub_BF0+F_j

ROM:0C07                 dec     b

ROM:0C08                 jr      nz, loc_BF3

ROM:0C0A                 ret

ROM:0C0A ; End of function sub_BF0

ROM:0C0A

ROM:0C0A ; ---------------------------------------------------------------------------

È tardi, ho sonno e non ho voglia di commentare opcode per opcode quindi vi darò una spiegazione di come funziona e poi starà a voi studiarvela.

Questa compressione funziona solo su dati particolarmente ripetitivi.

Come funziona:

Ci sono 3 byte di “controllo”.

I primi 2 byte, li analizza, parte dal primo byte analizzando dal bit 0 al bit 7, se il bit è 1 copia il byte riferito dal puntatore che poi viene incrementato, se il bit è 0 viene copiato il 3° byte di “controllo” cioè quello che si ripete più volte.

Se qualcuno si è chiesto a cosa serva la call 0680 (sub_680) che abbiamo riutilizzato nel nostro codice, non fa altro che aspettare la possibilità di scrivere in VRAM (quindi aspetta il VBLANK) niente di più niente di meno :)

Appendice B: Rom Address equation

Per ottenere la posizione nel banco dei dati che inseriamo nella rom:

posizionenellarom-($4000*n°del_banco_in_cui_si_trovano_i_dati)

il primissimo banco viene considerato "0" e l'osffset è sempre uguale, in memoria e nella rom.

per ottenere la posizione in memoria basta fare:

posizione nel banco+$4000

Per il contrario cioè da posizione in memoria a rom (per sapere dove si trovano i dati compressi x esempio):

(posizione nella memoria - $4000)+($4000*numerodelbanco)=posizione nella rom

fatto semplicemente e in pochi secondi, provate a farlo con gli offset relativi alle font e al nostro codice inserito ;)

Un saluto dal vostro ev0 aka DarkDreAm… l’originale ;)

ev0luzione@yahoo.it
…quello nella mail è uno zero (ev0…) non una o xD

Un saluto a Geo, Syxtem, Clomax, Mat, DarkSchneider e a tutti gli altri del forum emuita, traduzioni e shmup. Un grazie particolare a Syxtem, active85k e Clomax.

Un grazie a Ex-Goku per il beta-reading.

Un saluto a cronaldo7: visto cosa si può fare con un’hack in asm?
